

## 29. 1stOpt 曲线几何问题建模与求解

### 29.1 问题描述

如表29.1二维数据以及对应的图29.1和图29.2所示散点数据分布图，需要求解以下7个模型问题：

- 1) 用下列公式对该数据进行拟合计算，要求拟合曲线必须通过起始 A 与终点 B；
 
$$y = p_1 + p_2 \cdot (1 - \exp(p_3 \cdot x))$$
- 2) 如图 29.1, A 和 B 两点连成一直线，试求该直线与 AB 曲线间的最大垂直距离 CD（与 x 轴垂直）；
- 3) 如图 29.1, 试求出不规则图形 ACD 与 BCD 面积相等时的 CD 直线长度；
- 4) 如图 29.1, 试求 AD 弧线与 BD 弧线长度相等时的 CD 直线长度；
- 5) 如图 29.2, A 和 B 两点连成一直线，求曲线 ADB 到直线 AB 的最大垂直距离 CD；
- 6) 如图 29.2, 试求出不规则图形 ACD 与 BCD 面积相等时的 CD 直线长度；
- 7) 如图 29.2, 试求 AD 弧线与 BD 弧线长度相等时的 CD 直线长度；

表 29.1 x-y 二维数据

x	4,5,6,7,8,9,10,11,12
y	1.4,1.3,1.3,1.4,1.7,2.4,3.9,6.2,13.0




图 29.1 x-y 散点示意图[a]




图 29.2 x-y 散点示意图[b]

## 29.2 问题求解

所有问题都是基于直线AB与曲线ADB，因此先将其对应的方程公式分别求出是模型建立和求解的基础。

### 1) ADB曲线及AB直线拟合

ADB曲线拟合公式已给出，唯一特殊点是要求过首尾两点的约束，在1stOpt中可以很容易实现这种过点约束，有三两种方式，一种是直接加等式约束，第二种通过关键字“SubjectTo”定义等式约束，第三种则是通过关键字“PassPoint”定义曲线强制过点坐标约束，代码分别如下，所得结果均相同。

过点约束拟合代码：

形式一	<pre>Variable x,y; Function y = p1+p2*(1-exp(p3*x));       1.4 = p1+p2*(1-exp(p3*4));       13.0 = p1+p2*(1-exp(p3*12));  Data; x = [4,5,6,7,8,9,10,11,12]; y= [1.4,1.3,1.3,1.4,1.7,2.4,3.9,6.2,13.0];</pre>
形式二	<pre>SubjectTo 1.4 = p1+p2*(1-exp(p3*4)),           13.0 = p1+p2*(1-exp(p3*12));  Variable x,y; Function y = p1+p2*(1-exp(p3*x));  Data; x = [4,5,6,7,8,9,10,11,12]; y= [1.4,1.3,1.3,1.4,1.7,2.4,3.9,6.2,13.0];</pre>
形式三	<pre>PassPoint = [[4,1.4],[12,13.0]];  Variable x,y; Function y = p1+p2*(1-exp(p3*x));  Data; x = [4,5,6,7,8,9,10,11,12]; y= [1.4,1.3,1.3,1.4,1.7,2.4,3.9,6.2,13.0];</pre>

计算结果：

```
Sum Squared Error (SSE): 0.239267792512582
Root of Mean Square Error (RMSE): 0.163050023855183
Correlation Coef. (R): 0.999088863426232
R-Square: 0.99817855702232
Adjusted R-Square: 0.997571409363094
Determination Coef. (DC): 0.998014631460565
Chi-Square: 0.0457942529086824
F-Statistic: 1498.35730327023
Constrained Functions: 1.4-(p1+p2*(1-exp(p3*4))) = 3.62492258432212E-11
                      13.0-(p1+p2*(1-exp(p3*12))) = 4.35917968388821E-12
```

Parameter	Best Estimate
-----------	---------------

p1	1.38568073968642
p2	-0.000530999702303466
p3	0.832752797628872

可得 ADB 最终曲线方程为:

$$y = 1.38568073968642 - 0.000530999702303466 \cdot (1 - \exp(0.832752797628872 \cdot x))$$

AB直线拟合:

直线AB属于线性拟合, 相对更为简单, 也给出三种方式: 矩阵计算直接求解、线性拟合求解以及一般拟合求解, 代码及求解方式不同, 但结果均一致。

直线拟合代码:

矩阵计算直接求解	<pre>Constant x(2,2)=[1,4,                 1,12], y(2,1)=[1.4,         13]; f0=Matrix((X^T*X)^Inv*X^T*y);</pre>
线性拟合求解	<pre>Function y=LineFit(x); Data; 4,1.4 12,13</pre>
一般拟合求解	<pre>Function y=b0+b1*x; Data; 4,1.4 12,13</pre>

计算结果:

直接矩阵计算	<pre>f0(2,1) = [-4.4,             1.45];</pre>
线性及一般拟合	<pre>Sum Squared Error (SSE): 4.93038065763132E-30 Root of Mean Square Error (RMSE): 1.57009245868378E-15 Correlation Coef. (R): 1 R-Square: 1 Adjusted R-Square: 1 Determination Coef. (DC): 1 Chi-Square: 1.76085023486833E-30  Parameter      Best Estimate ----- b0              -4.4 b1              1.45</pre>

可得 AB 最终直线方程为:

$$y = -4.4 + 1.45 \cdot x$$

## 2) 最大垂直距离CD求解

参考图29.1, C和D点分别在直线AB和曲线ADC上, 因为垂直与x轴, 因此C和D两点的

横坐标x3与x4相等，只需用各自对应的纵坐标值相减即可得到CD线段长度。

代码：

---

```

Constant b1=-4.4, b2=1.45; //直线方程系数
Constant p1=1.385680739886, p2=-0.000530999691, p3=0.8327527994; //曲线方程系数
ConstStr y3=b1+b2*x, y4 = p1+p2*(1-exp(p3*x)); //两点垂直坐标
PassParameter y3, y4;
Parameter x=[4,12];
MaxFunction y3-y4;

```

---

结果：

---

```

Objective Function (Max.): 6.56933303628036
x: 9.72116949882341

PassParameter:
y3: 9.69569577329394
y4: 3.12636273701358

```

---

最大垂直距离CD为6.569333，对应的C和D两点坐标分别为[9.7211695, 9.695696]和[9.7211695, 3.126363]。

### 3) 不规则图形ACD与BCD面积相等时的CD直线长度

垂直线CD将AB直线与ADB曲线围成的闭环面积划分成了两部分即ACD和BCD两部分，ACD不规则面积可视作梯形ACFE面积减去AD段曲线与x轴围成的面积ADFE：

$$S_{ACD} = S_{ACFE} - S_{ADFE}$$

同理：

$$S_{BCD} = S_{BCFH} - S_{BDFH}$$

梯形ACFE和BCFH的面积非常容易计算，而曲线与x轴所围成的面积则需要使用积分计算，如ADFE和BDFH面积计算公式如下，注意此时C和D两点的横坐标x3与x4值是相同的：

$$S_{ADFE} = \int_{x=x_1}^{x=x_4} (p_1 + p_2 \cdot (1 - \exp(p_3 \cdot x))) dx$$

$$S_{BDFH} = \int_{x=x_4}^{x=x_2} (p_1 + p_2 \cdot (1 - \exp(p_3 \cdot x))) dx$$

代码：

---

```

Constant x1=4, y1=1.4, x2=12, y2=13; //A 和 B 两点坐标值
Constant b1=-4.4, b2=1.45; //直线方程系数
Constant p1=1.385680739886, p2=-0.000530999691, p3=0.8327527994; //曲线方程系数
ConstStr f1(v)=b1+b2*v, //直线方程
f2(v)=p1+p2*(1-exp(p3*v)), //曲线方程
SADFE=int(f2(x), x=x1, x4), //ADEF 不规则面积
SBDFH=int(f2(x), x=x4, x2), //BDFH 不规则面积
y3=f1(x4),
y4=f2(x4);
ConstStr SACFE=(y1+y3)*(x4-x1)/2, //ACFE 梯形面积
SBCFH=(y3+y2)*(x2-x4)/2, //BCFH 梯形面积

```

---

---

```

s1=SACFE-SADFE,
s2=SBCFH-SBDFH;
PassParameter y3, y4, s1, CD=y3-y4;
Parameter x4=[4,12];
Function s1=s2;

```

---

结果:

---

```

Objective Function (Min.): 5.24334427359425E-32
x4: 8.87624408638654

```

```

PassParameter:
y3: 8.47055392526048
y4: 2.24669127882107
s1: 16.2945490387131
cd: 6.22386264643941

```

---

上述结果表明， $x_3=x_4=8.876244$ 时，不规则面积ACD与BCD相等且为16.294549，此时CD长度为6.2238626，C、D两点坐标分别为[8.876244, 8.4705539]和[8.876244, 2.2466912]。

#### 4) AD弧线与BD弧线长度相等时的CD直线长度

此问题与前一问题唯一不同之处是将面积相等换成了弧线长度相等。曲线弧微分公式如下：

$$ds = \sqrt{1 + (y')^2} dx$$

弧长计算公式则为：

$$s = \int_{x=x_1}^{x=x_2} \sqrt{1 + (y')^2} dx$$

微分 $y'$ 在1stOpt模型语言中可用“Diff()”函数表示。

代码:

---

```

Constant x1=4,y1=1.4,x2=12,y2=13;           //A 和 B 两点坐标值
Constant b1=-4.4, b2=1.45;                 //直线方程系数
Constant p1=1.385680739886, p2=-0.000530999691, p3=0.8327527994; //曲线方程系数
ConstStr f1(v)=b1+b2*v,                    //直线方程
          f2(v)=p1+p2*(1-exp(p3*v)),        //曲线方程
          f3(v)=diff(f2(v),v),              //微分
          y3=f1(x4),
          y4=f2(x4),
          AD=int(sqrt(1+(f3(x))^2),x=x1,x4), //AD 弧线长
          BD=int(sqrt(1+(f3(x))^2),x=x4,x2); //BD 弧线长
PassParameter y3, y4, s1=AD, CD=y3-y4;
Parameter x4=[4,12];
Function AD=BD;

```

---

结果:

---

```

Objective Function (Min.): 2.83989925879564E-29
x4: 10.5683801288241

```

---

---

PassParameter:  
y3: 10.9241511867949  
y4: 4.9109203277016  
s1: 8.22964285537294  
cd: 6.01323085909334

---

当 $x_3=x_4=10.56838$ 时, AD与BD弧线长度相等为8.2296428, 此时CD长度为6.0132308, C、D两点坐标分别为[10.56838, 10.924151]和[10.56838, 4.9109203]。

### 5) 曲线ADB到直线AB的最大垂直距离CD

参考图29.2, 点C和D不在同一垂直线上, 即x横坐标3与x4不同, CD长度:

$$CD = \sqrt{(x_3 - x_4)^2 + (y_3 - y_4)^2}$$

约束条件是CD与AB垂直, 数学方面可通过CD和AB两直线的斜率互为负来表示, 因为AB直线方程已求出, 其斜率 $b_2=1.45$ , 因此CD直线的斜率为:

$$\frac{x_3 - x_4}{y_3 - y_4} = -b_2$$

代码:

---

```
Constant b1=-4.4, b2=1.45;
Constant p1=1.38568073988605, p2=-0.000530999691000306, p3=0.832752799401239;
ConstStr f1(v)=b1+b2*v,f2(v)=p1+p2*(1-exp(p3*v));
ConstStr y3=f1(x3),y4=f2(x4);
ConstStr CD=sqrt((x3-x4)^2+(y3-y4)^2); //垂直线距离
PassParameter y3,y4;
Parameter x(3:4)=[4,12];
MaxFunction CD;
(x4-x3)/(y4-y3)=-b2; //CD 与 AB 垂直
```

---

结果:

---

```
Objective Function (Max.): 3.72962686234608
x3: 6.65089299292485
x4: 9.72116951270127

PassParameter:
y3: 5.24379483974103
y4: 3.12636275713648
```

---

```
Constrained Functions:
1: (x4-x3)/(((1.38568073988605+((-0.000530999691000306))*(1-exp(0.832752799401239*(x4)))))-((( (-4.4)+1.45*(x3)))))-(-1.45) = 8.72635297355373E-14
```

---

曲线到直线AB的最大距离CD为3.7296268, 对应的C和D两点坐标分别为[6.65089299, 5.2437948]和[9.7211695, 3.1263627]。

### 6) 不规则图形ACD与BCD面积相等时的CD直线长度

曲线到AB直线的垂直线CD将AB直线与ADB曲线围成的闭环面积划分成了两部分即ACD和BCD两部分, ACD不规则面积S1可视为三角形ACD面积加上梯形ADFE面积再减去AD段曲线与x轴围成的面积ADFE\_1:

$$S1 = S_{ACD} + S_{ADFE} - S_{ADFE\_1}$$

同理：

$$S2 = S_{BCD} + S_{BDFH} - S_{BDFH\_1}$$

三角形ACD和BCD，梯形ADFE和BDFH面积都容易计算，而曲线与x轴所围成的面积需要使用积分计算，ADFE\_1和BDFH\_1面积计算公式如下：

$$S_{ADFE\_1} = \int_{x=x_1}^{x=x_4} (p_1 + p_2 \cdot (1 - \exp(p_3 \cdot x))) dx$$

$$S_{BDFH\_1} = \int_{x=x_4}^{x=x_2} (p_1 + p_2 \cdot (1 - \exp(p_3 \cdot x))) dx$$

代码：

---

```

Constant [x1,y1,x2,y2]=[4,1.4,12,13];
Constant b1=-4.4, b2=1.45;
Constant p1=1.38568073988605, p2=-0.000530999691000306, p3=0.832752799401239;
ConstStr f1(v)=b1+b2*v,f2(v)=p1+p2*(1-exp(p3*v));
ConstStr y3=f1(x3),y4=f2(x4);
ConstStr AE=y1, DF=y4, BH=y2, EF=x4-x1 FH=x2-x4,
    CD=sqrt((x3-x4)^2+(y3-y4)^2),           //CD 直线长度
    AC=sqrt((x3-x1)^2+(y3-y1)^2),           //AC 直线长度
    BC=sqrt((x3-x2)^2+(y3-y2)^2),           //BC 直线长度
    SACD=0.5*AC*CD,                          //ACD 三角形面积
    SBCD=0.5*BC*CD,                          //BCD 三角形面积
    SADFE=(AE+DF)*(x4-x1)/2,                 //ADFE 梯形面积
    SBDFH=(DF+BH)*(x2-x4)/2,                 //BDFH 梯形面积
    SADFE_1=int(f2(x),x=x1,x4),              //ADFE 不规则面积
    SBDFH_1=int(f2(x),x=x4,x2),              //BDFH 不规则面积
    s1=SACD+SADFE-SADFE_1,
    s2=SBCD+SBDFH-SBDFH_1;
PassParameter CD,s1,y3,y4;
Parameter x(3:4)=[4,12];
Function s1=s2;
(x4-x3)/(y4-y3)=-b2;                       //CD 与 AB 垂直

```

---

结果：

---

```

Objective Function (Min.): 3.39151376088829E-29
x3: 7.36644760776894
x4: 10.3174907710591

PassParameter:
cd: 3.58478781401453
s1: 16.2945490396995
y3: 6.28134903126496
y4: 4.24614684968555

```

---

即C和D两点坐标分别为[7.3664476, 6.281349]和[10.3174907, 4.2461468]时，CD长度为3.5847878，且垂直于AB，划分的ACD面积与BCD面积相同，均为16.294549。

7) AD弧线与BD弧线长度相等时的CD直线长度

参考前面问题4, 不同之处是直线CD必须与AB垂直, 因此C和D两点横坐标x3与x4不同, 均为待求参数。

代码:

```

Constant x1=4,y1=1.4,x2=12,y2=13;           //A 和 B 两点坐标值
Constant b1=-4.4, b2=1.45;                 //直线方程系数
Constant p1=1.385680739886, p2=-0.000530999691, p3=0.8327527994; //曲线方程系数
ConstStr f1(v)=b1+b2*v,                    //直线方程
          f2(v)=p1+p2*(1-exp(p3*v)),        //曲线方程
          f3(v)=diff(f2(v),v),              //弧微分
          AD=int(sqrt(1+(f3(x))^2),x=x1,x4), //AD 弧线长
          BD=int(sqrt(1+(f3(x))^2),x=x4,x2); //BD 弧线长

ConstStr y3=f1(x3),y4=f2(x4);
PassParameter s1=AD, CD=sqrt((x3-x4)^2+(y3-y4)^2) ,y3,y4;
Parameter x(3:4)=[4,12];
Function AD=BD;
          (x4-x3)/(y4-y3)=-b2;              //CD 与 AB 垂直

```

结果:

```

Objective Function (Min.): 8.78164175195528E-25
x3: 7.75800632142407
x4: 10.5683801303892

PassParameter:
s1: 8.22964286022753
cd: 3.41390932824233
y3: 6.8491091660649
y4: 4.91092033229697

```

即C和D坐标分别为[7.758006, 6.849109]和[10.568380, 4.9109203]时, CD长度为3.413909, 且垂直于AB, 划分的弧长为8.229642。

问题结果汇总:

除了第一个约束拟合问题外, 第二至第七个问题输出内容都相似, 汇总如下表格。

表29.2 问题2至问题7计算结果对比

问题编号	C 与 D 点坐标	CD 直线长度
2	[9.7211695, 9.6956957], [9.7211695, 3.1263627]	6.5693330
3	[8.8762441, 8.4705539], [8.8762441, 2.2466912]	6.2238626
4	[10.5683801, 10.9241512], [10.5683801, 4.9109203]	6.0132308
5	[6.6508929, 5.2437948], [9.7211695, 3.1263627]	3.7296268
6	[7.3664476, 6.2813490], [10.3174907, 4.2461468]	3.5847878
7	[7.758006, 6.8491092], [10.5683801, 4.9109203]	3.4139093



## 29.3 小结

本案例注意演示了四方面的内容：1) 过点约束非线性拟合；2) 直接矩阵计算求解线性拟合参数；3) 曲线面积的求解及作为约束的优化问题；4) 弧微分及曲线长度的计算及作为约束条件的优化问题。

通过这些实际问题的建模与求解，更能体验1stOpt简洁易懂的模型语言及高效的求解算法。